



Documento de Trabajo SECMCA-02-2022

CONSEJOS PARA EL USO EFICIENTE DE EXCEL

Randall Romero Aguilar

Secretaría Ejecutiva
San José, Costa Rica



Consejos para el uso eficiente de Excel

Randall Romero-Aguilar*
rromero@secmca.org

Resumen

A pesar de ser una herramienta muy útil y conocida, con Excel es fácil equivocarse cuando se trabaja en el manejo de datos. En este documento de trabajo se ofrecen doce consejos prácticos para desarrollar mejores hojas de cálculo, reduciendo lo más posible esos errores. Además, se ofrecen nuevas funciones para realizar cálculos comunes con series de tiempo, y se explica cómo desarrollar nuevas funciones sin necesidad de escribir macros en VBA.

Palabras clave: Excel, OFFSET, LET, LAMBDA, Power Query,

Clasificación JEL: C80, C88

1 ¿Por qué estos consejos para Excel?

Excel está en todas partes. En cualquier empresa, institución financiera, o agencia del gobierno que podamos imaginar encontraremos personas que utilizan Excel para almacenar, transformar, y visualizar datos.

Parte de la gran popularidad de Excel se debe a su flexibilidad y a la facilidad con la que se puede aprender a hacer tareas. Excel permite copiar-pegar datos, hacer gráficos simplemente con unos cuantos clics en algunos botones, introducir fórmulas con un conocimiento mínimo de programación (básicamente, saber cuál fórmula utilizar), y ver resultados de inmediato. Otra parte de su popularidad se debe a ser una suerte de "lengua franca" del mundo del análisis de datos: como millones de personas usan Excel, nuevos analistas de datos se ven en la necesidad de aprenderlo también.

Pero toda esa flexibilidad y facilidad de uso conlleva un costo enorme: es muy fácil equivocarse en una fórmula, o estropear una fórmula que funciona correctamente. Borrar una

*Economista Consultor de la Secretaría Ejecutiva del Consejo Monetario Centroamericano (SECMCA) y profesor de economía en la Universidad de Costa Rica (UCR). Doctor en Economía por la Ohio State University (OSU) y Master en Econometría por la London School of Economics (LSE). Las opiniones expresadas son las del autor y no necesariamente representan la posición de la SECMCA, ni de los miembros del CMCA.

celda por aquí, eliminar una fila por allá, cambiar el nombre de un archivo vinculado, modificar accidentalmente el contenido de una celda, son solo algunos ejemplos de cómo se introducen errores en las hojas de cálculo. Lo más grave de este asunto es que en muchísimos casos es difícil detectar estos errores.

El costo de tales errores puede ser enorme. En Internet abundan historias de episodios en los que el mal uso de una hoja de cálculo contribuyó a serios problemas. Por citar algunos ejemplos:

- En la debacle conocida como "London Whale", en la cual JPMorgan Chase & Co. perdió al menos US\$ 6200 millones en 2012, se identificó que parte del problema se suscitó por un manejo inadecuado de datos, los cuales se copiaban y pegaban manualmente entre distintas hojas de cálculo, así como por errores en hojas de cálculo que subestimaban los riesgos de las operaciones. (Hurtado [2016](#); Lopez [2013](#))
- En un influyente artículo, Reinhart y Rogoff ([2010](#)) encontraron evidencia estadística según la cual la tasa de crecimiento económico desciende en 0.1% cuando el nivel de deuda de un país supera el 90% de su PIB. Este hallazgo fue citado numerosamente por políticos que buscaban justificar políticas de austeridad fiscal. No obstante, tres años después Herndon, Ash y Pollin ([2013](#)) encontraron que en el análisis original, llevado a cabo en Excel, por error se había excluido a cinco países de la muestra; al corregir este error de fórmula, la "caída de 0.1%" se convirtió en un aumento de 2.2% en el crecimiento económico (Borwein y Bailey [2013](#)).
- Al inicio de la pandemia del Covid-19, en Inglaterra se sub-reportaron cerca de 16 mil casos de la enfermedad por un error con el *formato* de los archivos en los que se guardaban los datos. Inicialmente estos datos se almacenaban en archivos CSV (datos separados por coma), y luego se convertían a Excel. El error se produjo cuando al hacer la conversión se utilizó el formato XLS de Excel en vez del más moderno XLSX, disponible desde 2007. El formato XLS solamente tiene $2^{16} = 65\,536$ filas y $2^8 = 256$ columnas, comparado con las $2^{20} = 1\,048\,576$ filas y $2^{14} = 16\,384$ columnas disponibles en el formato XLSX. Así, cuando se convertían archivos CSV a XLS, Excel simplemente truncaba el número de filas, perdiéndose así muchos de los registros.

Puesto que difícilmente Excel desaparecerá de la "caja de herramientas" de los analistas de datos, y que es de humanos equivocarse, en este documento de trabajo presento doce consejos para minimizar la incidencia de errores en nuestros libros de trabajo. Estos consejos están encaminados a **simplificar** nuestras hojas de cálculo, de manera que resulte lo más sencillo posible para nosotros mismos así como para nuestros colegas detectar posibles fallas en nuestras hojas, para así poder corregirlas oportunamente.



En algunos de los consejos se sugiere el uso de algunas funciones de Excel; por razones de espacio no explico detalladamente todos los alcances de estas funciones, aunque en tales casos encontrará párrafos, como éste, en los que aparece el icono de Excel: estos iconos son hipervínculos que abren páginas de documentación de Microsoft acerca de esas funciones específicas. En particular, el icono de este párrafo abre la página principal de ayuda para Excel.

Para esta nota, he escrito el nombre de las funciones de Excel tal como aparece en la versión en inglés del programa, pero la documentación vinculada está en español. Además, en el apéndice encontrará una tabla con las principales funciones discutidas en este documento, su nombre en la versión en español de Excel, así como una breve descripción.

Es importante señalar que algunas de las funciones recomendadas en este documento solamente están disponibles en la versión más reciente de Excel, la que viene incluida en *Microsoft 365*. Este es el caso con las funciones **LET**, **LAMBDA**, y **XLOOKUP**, cuyo uso recomiendo en los consejos 6 al 8. Bajo el sistema de distribución actual de Microsoft para sus programas de oficina (los que anteriormente se denominaban *Microsoft Office*, ahora *Microsoft 365*), estos programas reciben actualizaciones y mejoras continuas, usualmente una vez por mes. Es así como fueron introducidas las funciones mencionadas anteriormente.



Para ilustrar los primeros nueve consejos he preparado un libro de Excel que puede descargarse en el sitio <http://www.secmca.org/wp-content/uploads/2022/06/Consejos-para-el-uso-eficiente-de-Excel.xlsx>, o bien con el icono a la izquierda de este párrafo.

2 Los consejos

Estos son los doce consejos:

1: Evitemos las referencias a celdas	4
2: Aprovechemos las fórmulas de matrices	5
3: Organizemos los datos en tablas	6
4: ¡Cuidémonos de las celdas sin números!	7
5: Seleccionemos rangos de datos con OFFSET	8
6: Simplifiquemos funciones con LET	9
7: Definamos nuevas funciones con LAMBDA	10
8: Usemos XLOOKUP para leer datos de tablas	11
9: Especifiquemos fechas como números	12
10: Compartamos datos con Power Query	13
11: Transformemos datos con Power Query	15
12: Complementemos Excel con otras herramientas	18

Consejo N° 1: Evitemos las referencias a celdas

La forma más usual en Excel de expresar los argumentos de una fórmula es por medio de una referencia a las celdas que contienen los valores que desean utilizarse en la fórmula. A pesar de que resulta sencillo utilizar estas referencias, en archivos de Excel que contienen numerosas fórmulas resulta difícil revisar que todas ellas sean correctas. Por ejemplo, considere esta fórmula, que aparece en la celda **E3** de la figura 1:


`=SUMPRODUCT(B3:B6,C3:C6)`

De entrada, sabemos que corresponde al cálculo de un producto punto de dos vectores, en este caso delimitados por los rangos de celdas **\$B\$3:\$B\$6** y **\$C\$3:\$C\$6**, pero para saber a qué se refieren exactamente esas celdas nos vemos obligados a navegar en la hoja de cálculo hasta esas celdas y ver su contenido y las celdas circundantes en búsqueda de etiquetas que las identifiquen (en este caso *cantidad* y *precio* en las celdas **B2** y **C2**). Aunque esto es muy sencillo en nuestro pequeño ejemplo, en hojas de cálculo muy grandes esta tarea se vuelve muy fastidiosa y consume mucho tiempo. En contraste, consideremos la fórmula de la celda **E4** de la figura 1:

`=SUMPRODUCT(cantidades,precios)`

En este caso hemos realizado exactamente el mismo cálculo, utilizando los mismos datos, pero refiriéndonos a ellos con **nombres**, en este caso *cantidades* y *precios*. Gracias a estos nombres, es más sencillo adivinar que la fórmula calcula el valor de un conjunto de bienes o de las ventas o producción en un período determinado.

Para asignarle un nombre a un rango de celdas, seleccionamos las celdas y escribimos el nombre deseado en el cuadro de *Nombre* (en la figura 1 es el espacio donde se lee "E3" en la parte superior izquierda). O bien, con **Formulas → Defined Names → Name Manager**.



	A	B	C	D	E
1					
2	trimestre	cantidad	precio		TOTAL VENDIDO
3	2020Q1	5	1		=SUMPRODUCT(\$B\$3:\$B\$6,\$C\$3:\$C\$6)
4	2020Q2	8	2		=SUMPRODUCT(cantidades,precios)
5	2020Q3	3	3		
6	2020Q4	7	4		
7					

Figura 1: Referencia a celdas versus nombres de rangos

Consejo N° 2: Aprovechemos las fórmulas de matrices

A menudo necesitamos calcular la misma fórmula para distintos datos. Comúnmente, lo hacemos escribiendo la fórmula requerida para el primer grupo de datos, y luego simplemente copiamos la fórmula a las demás celdas, aprovechando las referencias relativas de Excel. Un ejemplo de ello lo vemos en la figura 2, donde hemos calculado el producto de cantidad (columna B) y precio (columna C), usando las fórmulas que aparecen en las celdas E3:E6, a saber =B3*C3, =B4*C4, =B5*C5, =B6*C6. Este procedimiento está muy propenso a errores (es difícil detectar cuando se copia mal una celda) y es difícil de revisar, puesto que es necesario revisar una fórmula individual para cada dato calculado.



Una mejor alternativa es que usemos las fórmulas de matriz, que operan elemento por elemento y permiten realizar muchos cálculos repetidos con una única fórmula. En nuestro ejemplo anterior, esto lo implementamos en la celda F3, =cantidades*precios, donde hemos aprovechado además los nombres que les hemos asignado en el consejo anterior a los rangos de datos que deseamos multiplicar.



Lo más conveniente de este procedimiento es que sólo escribimos la fórmula en la primera celda, y Excel automáticamente *desborda* los resultados en celdas adyacentes, sin necesidad de seleccionarlás previamente. Es importante señalar que, para que Excel pueda desbordar (completar los cálculos de todas las celdas) es necesario que las celdas de destino (F3:F6 en la imagen) estén vacías.

Un detalle interesante de las fórmulas de matriz es que sus resultados pueden ser utilizados directamente por otras fórmulas, sin necesidad de guardarlos en un rango de celdas. Por ejemplo, si quisiéramos obtener la suma de la columna "total", sin necesidad de generar explícitamente esa columna, simplemente podemos anidar las funciones y escribir:

=SUM(cantidades*precios)

	A	B	C	D	E	F	G
1							
2	trimestre	cantidad	precio	total	fórmula1	fórmula2	
3	2020Q1	5	1	5	=B3*C3	=cantidades*precios	
4	2020Q2	8	2	16	=B4*C4		
5	2020Q3	3	3	9	=B5*C5		
6	2020Q4	7	4	28	=B6*C6		
7							

Figura 2: Fórmulas individuales versus fórmulas de matrices

Consejo N° 3: Organizemos los datos en tablas

Una vez que comenzamos a trabajar con enormes cantidades de datos, posiblemente no será práctico ir por el libro de Excel asignando nombres de celdas. Por fortuna, Excel permite organizar los datos en *tablas*, para las cuales automáticamente crea nombres para cada columna a partir de los encabezados de la primera fila. Una vez creada la tabla, cualquier fórmula que escribamos en alguna columna se *desbordará* a las demás celdas de la columna, como una función de matriz dinámica. Además, a diferencia de los nombres que asignamos manualmente (consejo 1) si posteriormente añadimos más observaciones al final de la tabla Excel automáticamente expandirá la tabla para que incluya estos nuevos registros, actualizando además todas las columnas calculadas de la tabla.

Un ejemplo se presenta en la figura 3, donde hemos definido el rango **A2:D6** como una tabla. Vemos que para calcular la columna "total" multiplicamos las columnas "cantidad" y "precio": `=[@cantidad]*[@precio]`. A pesar de que en la figura se observa la fórmula en todas las celdas de la columna, solo es necesario escribirla en una de ellas y Excel completa el resto. Incluso, al editar fórmulas en tablas Excel automáticamente utiliza los nombres de las columnas, en vez de las referencias de celda, facilitando mucho la labor.



Para crear una tabla de Excel, seleccionamos cualquier celda de los datos y luego **Insert → Tables → Table**. Por defecto, Excel le asigna a la tabla un nombre genérico como "Table1", pero es recomendable cambiarlo por uno que describa los datos, seleccionando alguna celda de la tabla y luego **Table Design → Properties → Table Name:**.

Una de las muchas ventajas de las tablas es que pueden ser referidas desde fuera de la tabla, en otras hojas o incluso en otros archivos de Excel. Por ejemplo, la tabla de la figura la hemos llamado "Ventas", por lo que si quisiéramos la suma de los valores de la columna "total" lo podemos obtener con `=SUM(Ventas[total])` como vemos en A8.

	A	B	C	D
1				
2	trimestre	cantidad	precio	total
3	2020Q1	5	1	<code>=[@cantidad]*[@precio]</code>
4	2020Q2	8	2	<code>=[@cantidad]*[@precio]</code>
5	2020Q3	3	3	<code>=[@cantidad]*[@precio]</code>
6	2020Q4	7	4	<code>=[@cantidad]*[@precio]</code>
7				
8	<code>=SUM(Ventas[total])</code>			

Figura 3: Trabajando con tablas

Consejo N° 4: ¡Cuidémonos de las celdas sin números!

Una fuente común de errores al manipular datos con Excel es el tratamiento que este programa le da a las celdas no numéricas a la hora de evaluar fórmulas: en algunos casos, las omite de los cálculos, y en otras las considera como si tuviesen el valor cero.

Para ilustrar esto, en la figura 4a tenemos datos de un indicador ficticio para los primeros nueve meses de 2022. Deseamos suavizar ese indicador con una media móvil centrada, en particular: $\bar{y}_t = \frac{1}{3}(y_{t-1} + y_t + y_{t+1})$, donde y_t se refiere al indicador en el mes t . En primer instancia, en **C2** introducimos `=AVERAGE(B1:B3)` y copiamos la fórmula hasta el final de la columna (celda **C10**). El problema lo tenemos precisamente en los extremos de la columna: en **C2**, al evaluar `=AVERAGE(B1:B3)` Excel da por resultado el promedio de **B2:B3** porque en **B1** hay un texto ("indicador", el nombre de la columna). Similarmente, en **C10**, al evaluar `=AVERAGE(B9:B11)` el resultado es el promedio de **B9:B10**, porque la celda **B11** está vacía. Una forma rápida de arreglar este problema es borrar las celdas **C2** y **C10**, pero al actualizar la tabla con nuevos datos (por ejemplo, al disponer de datos para octubre 2022) posiblemente reintroduciremos el error al copiar la fórmula hacia abajo.

Para corregirlo, usaremos funciones de Excel para verificar que todas las celdas (**AND**) requeridas tengan datos numéricos (**ISNUMBER**); de ser así (**IF**), realizamos el cálculo, de lo contrario regresamos un valor nulo (un texto vacío). Veamos entonces la fórmula en **D2**:

`=IF(AND(ISNUMBER(B1:B3)), AVERAGE(B1:B3), "")`

Con esta nueva fórmula, hemos logrado eliminar los datos incorrectos en los extremos de la serie: vemos que en **D2** y **D10** no hay valor, a pesar de que la fórmula sí está presente.

El lector habrá notado, eso sí, que en estas fórmulas hemos hecho caso omiso a uno de los consejos: usamos referencias de celda, en vez de nombres. Esto lo corregiremos en la siguiente sección.

	A	B	C	D
1	mes	indicador	MM(3), mal	MM(3), bien
2	2022-Jan	4	6.50	
3	2022-Feb	9	6.67	6.67
4	2022-Mar	7	8.33	8.33
5	2022-Apr	9	7.33	7.33
6	2022-May	6	8.00	8.00
7	2022-Jun	9	6.00	6.00
8	2022-Jul	3	5.33	5.33
9	2022-Aug	4	5.00	5.00
10	2022-Sep	8	6.00	
11				

(a) Los valores

	B	C	D
1	indicador	MM(3), mal	MM(3), bien
2	4	=AVERAGE(B1:B3)	=IF(AND(ISNUMBER(B1:B3)),AVERAGE(B1:B3),"")
3	9	=AVERAGE(B2:B4)	=IF(AND(ISNUMBER(B2:B4)),AVERAGE(B2:B4),"")
4	7	=AVERAGE(B3:B5)	=IF(AND(ISNUMBER(B3:B5)),AVERAGE(B3:B5),"")
5	9	=AVERAGE(B4:B6)	=IF(AND(ISNUMBER(B4:B6)),AVERAGE(B4:B6),"")
6	6	=AVERAGE(B5:B7)	=IF(AND(ISNUMBER(B5:B7)),AVERAGE(B5:B7),"")
7	9	=AVERAGE(B6:B8)	=IF(AND(ISNUMBER(B6:B8)),AVERAGE(B6:B8),"")
8	3	=AVERAGE(B7:B9)	=IF(AND(ISNUMBER(B7:B9)),AVERAGE(B7:B9),"")
9	4	=AVERAGE(B8:B10)	=IF(AND(ISNUMBER(B8:B10)),AVERAGE(B8:B10),"")
10	8	=AVERAGE(B9:B11)	=IF(AND(ISNUMBER(B9:B11)),AVERAGE(B9:B11),"")
11			

(b) Las fórmulas

Figura 4: Fórmulas que incluyen celdas que carecen de datos

Consejo N° 5: Seleccionemos rangos de datos con **OFFSET**



Cuando necesitamos trabajar con rangos de celdas, que cambian conforme cambiamos de celda, no es práctico definir nombres para cada posible rango. Además, no podemos aprovechar la facilidad de los nombres generados automáticamente por las tablas de Excel porque en tales tablas las fórmulas operan por fila: el carácter **@** de las fórmulas de la columna D en la figura 3 significa "de esta fila". Esto lo podemos resolver con la función **OFFSET**, la cual da por resultado un rango de celdas relativa a la posición de una celda en particular.

En el ejemplo de la sección anterior, para la fórmula de la celda **D2** necesitamos datos de la columna B entre las filas 1 y 3, por ello la referencia a **B1:B3**. Otra manera de referirnos a estos datos, tomando como punto de partida a la celda **B2** es **=OFFSET(B2,-1,0,3,1)**, que nos dice que deseamos el rango de celdas que empieza una fila arriba (-1) y en la misma columna(0) que la celda **B2**, y contiene un total de tres filas (3) y una columna (1). Gracias a que ahora solo hacemos referencia explícita a una celda, y tal celda ocupa la misma fila que la celda donde queremos incluir la fórmula, podemos de nuevo usar los nombres de tabla. Por ejemplo, en la celda **E2** de la figura 5 hemos introducido la fórmula

```
=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),  
    AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
```

la cual ahora utiliza la referencia a la columna **[@indicador]**. Por tratarse de una tabla, Excel autocompleta el resto de la columna utilizando exactamente la misma fórmula.

OFFSET también es muy útil cuando necesitamos realizar cálculos con celdas separadas entre sí por una cantidad fija de filas y/o columnas. Por ejemplo, supongamos que tenemos datos mensuales del IPC en una tabla; para calcular la tasa de inflación podemos usar

```
=100 * ([@IPC] / OFFSET([@IPC],-12,0) - 1)
```

aunque, de nuevo, necesitaríamos ajustar esta fórmula para lidiar con las celdas que no contienen números.

	B	D	E
1	indicador	MM(3),bien	MM(3), mejor
2	4	=IF(AND(ISNUMBER(B1:B3)),AVERAGE(B1:B3),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
3	9	=IF(AND(ISNUMBER(B2:B4)),AVERAGE(B2:B4),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
4	7	=IF(AND(ISNUMBER(B3:B5)),AVERAGE(B3:B5),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
8	3	=IF(AND(ISNUMBER(B7:B9)),AVERAGE(B7:B9),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
9	4	=IF(AND(ISNUMBER(B8:B10)),AVERAGE(B8:B10),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
10	8	=IF(AND(ISNUMBER(B9:B11)),AVERAGE(B9:B11),"")	=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
11			

Figura 5: Usando la función **OFFSET** para referirnos a rangos

Consejo N° 6: Simplifiquemos funciones con LET

Cuando empezamos a tener cuidado de verificar que trabajamos con celdas numéricas hemos introducido un nuevo inconveniente: ¡las fórmulas se hacen cada vez más grandes e ilegibles! Veamos de nuevo la fórmula del consejo anterior:

```
=IF(AND(ISNUMBER(OFFSET([@indicador],-1,0,3,1))),  
    AVERAGE(OFFSET([@indicador],-1,0,3,1)),"")
```

Vemos que hemos usado dos veces la expresión `OFFSET([@indicador],-1,0,3,1)`. Para simplificar esta fórmula, Excel ofrece la función **LET**, que permite darle nombre a expresiones intermedias dentro de una fórmula. Así, el cálculo anterior lo podemos obtener como

```
=LET(datos, OFFSET([@indicador],-1,0,3,1),  
    IF(AND(ISNUMBER(datos)), AVERAGE(datos),""))
```

Podemos definir tantos nombres como sean necesarios (máximo 126). Por ejemplo, la fórmula anterior es más legible como aparece en la celda **F2** de la figura 6:

```
=LET(datos, OFFSET([@indicador],-1,0,3,1),  
    no_faltan_datos, AND(ISNUMBER(datos)),  
    IF(no_faltan_datos, AVERAGE(datos),""))
```

Vemos que la última expresión de la fórmula nos dice de qué se trata: si no faltan datos, calcule su promedio.

Antes de terminar, una pequeña advertencia: al escoger los nombres que le damos a cálculos intermedios, debemos evitar nombres que ya existen en Excel, como "y1" porque serían confundidos con la referencia a la celda Y1, o bien "sum" porque se confundiría con la función **SUM**.

	B	F
1	indicador	MM(3), mejor2
2	4	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))
3	9	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))
4	7	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))
8	3	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))
9	4	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))
10	8	=LET(datos, OFFSET([@indicador],-1,0,3,1), no_faltan_datos,AND(ISNUMBER(datos)), IF(no_faltan_datos, AVERAGE(datos),""))

Figura 6: Usando la función LET para simplificar fórmulas

Consejo N° 7: Definamos nuevas funciones con LAMBDA



En el consejo anterior, usando **LET** logramos expresar una fórmula compleja de manera más legible, pero ciertamente terminamos con una fórmula más larga: ¡pasamos de 91 a 127 caracteres! Para trabajar con fórmulas más pequeñas, una opción (no muy atractiva) es ir guardando los resultados intermedios en celdas, y luego hacer varias fórmulas encadenando los insumos intermedios. Una mejor opción es usar la función **LAMBDA**, que permite definir nuevas funciones sin necesidad de usar VBA ni de definir macros. Para definir una nueva función **LAMBDA** abrimos el cuadro de diálogo de la figura 7a con **Formulas** → **Name Manager** → **Define Name**, asignamos un nombre en **Name:**, opcionalmente (pero muy recomendado) describimos la función en **Comment**, y en **Refers to:** escribimos la función.


Para definir una nueva función con **LAMBDA** escribimos una serie de expresiones, la última de ellas es el valor que devolverá la función y todas las anteriores son los nombres de los parámetros que toma la función. Por ejemplo, para el caso de nuestro ejemplo anterior, en el que calculamos una media móvil teniendo cuidado de manipular adecuadamente las celdas no numéricas, hemos llamado a la nueva fórmula **mi_promedio**, que toma por argumento una celda que llamamos "x" y calcula lo mismo que la última fórmula del consejo anterior (ver figura 7a)

```
=LAMBDA(x, LET(datos, OFFSET(x, -1, 0, 3, 1), no_faltan_datos, AND(ISNUMBER(datos))), IF(no_faltan_datos, AVERAGE(datos), "")))
```

Hecho esto, podemos utilizar nuestra nueva función como cualquier otra función de Excel. Por ejemplo, en la celda **G2** en la figura 7b ahora simplemente calculamos el promedio ponderado con

```
=mi_promedio([@indicador])
```

donde el argumento "x" toma el valor **[@indicador]**, que significa "el valor de la columna "indicador" en la misma fila "@" que está esta fórmula, de la tabla actual.



(a) Definiendo una función con LAMBDA

	B	G
1	indicador	MM(3), mejor3
2	4	=mi_promedio([@indicador])
3	9	=mi_promedio([@indicador])
4	3	=mi_promedio([@indicador])
5	3	=mi_promedio([@indicador])
6	3	=mi_promedio([@indicador])
7	3	=mi_promedio([@indicador])
8	3	=mi_promedio([@indicador])
9	4	=mi_promedio([@indicador])
10	8	=mi_promedio([@indicador])
11		

(b) Usando la nueva función

Figura 7: Simplificando nuestros cálculos con funciones LAMBDA

Consejo N° 8: Usemos **XLOOKUP** para leer datos de tablas



Muchas veces deseamos construir un cuadro con datos que posiblemente estén en lugares distintos de nuestro libro de cálculo. Como mencionamos en el primer consejo, no es recomendable hacer esto con referencias de celda. Cuando los datos de origen los hemos guardado como tablas (siguiendo el consejo 3) podemos hacer el cuadro utilizando **XLOOKUP**.

Por ejemplo, supongamos que tenemos una tabla con datos de la variación anual media del IMAE de los países de CARD (ver figura 8) calculada, como no, con una función lambda llamada **variacion_acumulada** a partir de datos de otra tabla que contiene el nivel del IMAE para estos países, y que deseamos construir una tabla resumen con los datos de Guatemala y El Salvador de los últimos tres años. Esto lo logramos escribiendo en **AB17**

```
=XLOOKUP([@Año], IMAE_VA[Mes], IMAE_VA[Guatemala], "")
```

y una fórmula similar para El Salvador en **AC17**. La función **XLOOKUP** busca el primer argumento (el año de la celda AA17, **[@Año]**) en el vector del segundo argumento (la columna "Mes" de la tabla "IMAE_VA", **IMAE_VA[Mes]**): si lo encuentra, devuelve la entrada correspondiente en el vector siguiente (el dato de "Guatemala" en la tabla "IMAE_VA", **IMAE_VA[Guatemala]**), de lo contrario devuelve una celda en blanco con "".

La principal ventaja de **XLOOKUP** sobre la más conocida **VLOOKUP** es su flexibilidad y que es más explícita: con **VLOOKUP** tendríamos que indicar la columna deseada con un número de posición, lo que dificulta revisar la fórmula.

▲	Mes	Costa Rica	El Salvador	Guatemala	Honduras	Nicaragua	República Dominicana	Z	AA	AB	AC
11	2007-Oct										
12	2007-Nov										
13	2007-Dec										
14	2008-Jan	5.81	4.62	4.77	5.34	7.76	7.55				
15	2008-Feb	5.71	4.73	4.84	5.69	7.74	7.52				
16	2008-Mar	5.56	5.21	4.88	5.79	7.54	7.45				
17	2008-Apr	5.37	5.40	4.86	6.07	7.23	7.29				
18	2008-May	5.15	4.85	4.81	6.16	6.86	6.98				
19	2008-Jun	4.92	3.94	4.73	6.19	6.47	6.50				
20	2008-Jul	4.69	3.18	4.62	6.02	6.08	5.87				
21	2008-Aug	4.46	2.82	4.49	5.82	5.69	5.65				
22	2008-Sep	4.23	2.76	4.26	5.72	5.49	5.40				
23	2008-Oct	4.00	2.76	4.03	5.52	5.29	5.20				
24	2008-Nov	3.77	2.76	3.80	5.33	5.06	5.00				
25	2008-Dec	3.54	2.76	3.57	5.14	4.83	4.80				
26	2009-Jan	3.31	2.76	3.34	4.95	4.60	4.60				
27	2009-Feb	3.08	2.76	3.11	4.76	4.43	4.40				
28	2009-Mar	2.85	2.76	2.88	4.57	4.24	4.20				
29	2009-Apr	2.62	2.76	2.65	4.38	4.05	4.00				
30	2009-May	2.39	2.76	2.42	4.19	3.86	3.80				
31	2009-Jun	2.16	2.76	2.19	4.00	3.67	3.60				
32	2009-Jul	1.93	2.76	1.96	3.81	3.48	3.40				
33	2009-Aug	1.70	2.76	1.73	3.62	3.29	3.20				
34	2009-Sep	1.47	2.76	1.50	3.43	3.10	3.00				
35	2009-Oct	1.24	2.76	1.27	3.24	2.91	2.80				
36	2009-Nov	1.01	2.76	1.04	3.05	2.72	2.60				
37	2009-Dec	0.78	2.76	0.81	2.86	2.53	2.40				
38	2010-Jan	0.55	2.76	0.58	2.67	2.34	2.20				
39	2010-Feb	0.32	2.76	0.35	2.48	2.15	2.00				
40	2010-Mar	0.09	2.76	0.12	2.29	1.96	1.80				
41	2010-Apr	-0.14	2.76	-0.11	2.10	1.77	1.60				
42	2010-May	-0.37	2.76	-0.34	1.91	1.58	1.40				
43	2010-Jun	-0.60	2.76	-0.57	1.72	1.39	1.20				
44	2010-Jul	-0.83	2.76	-0.80	1.53	1.20	1.00				
45	2010-Aug	-1.06	2.76	-1.03	1.34	1.01	0.80				
46	2010-Sep	-1.29	2.76	-1.26	1.15	0.82	0.60				
47	2010-Oct	-1.52	2.76	-1.49	0.96	0.63	0.40				
48	2010-Nov	-1.75	2.76	-1.72	0.77	0.44	0.20				
49	2010-Dec	-1.98	2.76	-1.95	0.58	0.25	0.00				
50	2011-Jan	-2.21	2.76	-2.18	0.39	0.06	-0.20				
51	2011-Feb	-2.44	2.76	-2.41	0.20	-0.13	-0.40				
52	2011-Mar	-2.67	2.76	-2.64	0.01	-0.34	-0.60				
53	2011-Apr	-2.90	2.76	-2.87	-0.18	-0.55	-0.80				
54	2011-May	-3.13	2.76	-3.10	-0.39	-0.76	-1.00				
55	2011-Jun	-3.36	2.76	-3.33	-0.60	-0.97	-1.20				
56	2011-Jul	-3.59	2.76	-3.56	-0.81	-1.18	-1.40				
57	2011-Aug	-3.82	2.76	-3.79	-1.02	-1.39	-1.60				
58	2011-Sep	-4.05	2.76	-4.02	-1.23	-1.60	-1.80				
59	2011-Oct	-4.28	2.76	-4.25	-1.44	-1.81	-2.00				
60	2011-Nov	-4.51	2.76	-4.48	-1.65	-2.02	-2.20				
61	2011-Dec	-4.74	2.76	-4.71	-1.86	-2.23	-2.40				
62	2012-Jan	-4.97	2.76	-4.94	-2.07	-2.44	-2.60				
63	2012-Feb	-5.20	2.76	-5.17	-2.28	-2.65	-2.80				
64	2012-Mar	-5.43	2.76	-5.40	-2.49	-2.86	-3.00				
65	2012-Apr	-5.66	2.76	-5.63	-2.70	-3.07	-3.20				
66	2012-May	-5.89	2.76	-5.86	-2.91	-3.28	-3.40				
67	2012-Jun	-6.12	2.76	-6.09	-3.12	-3.49	-3.60				
68	2012-Jul	-6.35	2.76	-6.32	-3.33	-3.70	-3.80				
69	2012-Aug	-6.58	2.76	-6.55	-3.54	-3.91	-4.00				
70	2012-Sep	-6.81	2.76	-6.78	-3.75	-4.12	-4.20				
71	2012-Oct	-7.04	2.76	-7.01	-3.96	-4.33	-4.40				
72	2012-Nov	-7.27	2.76	-7.24	-4.17	-4.54	-4.60				
73	2012-Dec	-7.50	2.76	-7.47	-4.38	-4.75	-4.80				
74	2013-Jan	-7.73	2.76	-7.70	-4.59	-4.96	-5.00				
75	2013-Feb	-7.96	2.76	-7.93	-4.80	-5.17	-5.20				
76	2013-Mar	-8.19	2.76	-8.16	-5.01	-5.38	-5.40				
77	2013-Apr	-8.42	2.76	-8.39	-5.22	-5.59	-5.60				
78	2013-May	-8.65	2.76	-8.62	-5.43	-5.80	-5.80				
79	2013-Jun	-8.88	2.76	-8.85	-5.64	-6.01	-6.00				
80	2013-Jul	-9.11	2.76	-9.08	-5.85	-6.22	-6.20				
81	2013-Aug	-9.34	2.76	-9.31	-6.06	-6.43	-6.40				
82	2013-Sep	-9.57	2.76	-9.54	-6.27	-6.64	-6.60				
83	2013-Oct	-9.80	2.76	-9.77	-6.48	-6.85	-6.80				
84	2013-Nov	-10.03	2.76	-10.00	-6.69	-7.06	-7.00				
85	2013-Dec	-10.26	2.76	-10.23	-6.90	-7.27	-7.20				
86	2014-Jan	-10.49	2.76	-10.46	-7.11	-7.48	-7.40				
87	2014-Feb	-10.72	2.76	-10.69	-7.32	-7.69	-7.60				
88	2014-Mar	-10.95	2.76	-10.92	-7.53	-7.90	-7.80				
89	2014-Apr	-11.18	2.76	-11.15	-7.74	-8.11	-8.00				
90	2014-May	-11.41	2.76	-11.38	-7.95	-8.32	-8.20				
91	2014-Jun	-11.64	2.76	-11.61	-8.16	-8.53	-8.40				
92	2014-Jul	-11.87	2.76	-11.84	-8.37	-8.74	-8.60				
93	2014-Aug	-12.10	2.76	-12.07	-8.58	-8.95	-8.80				
94	2014-Sep	-12.33	2.76	-12.30	-8.79	-9.16	-9.00				
95	2014-Oct	-12.56	2.76	-12.53	-9.00	-9.37	-9.20				
96	2014-Nov	-12.79	2.76	-12.76	-9.21	-9.58	-9.40				
97	2014-Dec	-13.02	2.76	-12.99	-9.42	-9.79	-9.60				
98	2015-Jan	-13.25	2.76	-13.22	-9.63	-10.00	-9.80				
99	2015-Feb	-13.48	2.76	-13.45	-9.84	-10.21	-10.00				
100	2015-Mar	-13.71	2.76	-13.68	-10.05	-10.42	-10.20				
101	2015-Apr	-13.94	2.76	-13.91	-10.26	-10.63	-10.40				
102	2015-May	-14.17	2.76	-14.14	-10.47	-10.84	-10.60				
103	2015-Jun	-14.40	2.76	-14.37	-10.68	-11.05	-10.80				
104	2015-Jul	-14.63	2.76	-14.60	-10.89	-11.26	-11.00				
105	2015-Aug	-14.86	2.76	-14.83	-11.10	-11.47	-11.20				
106	2015-Sep	-15.09	2.76	-15.06	-11.31	-11.68	-11.40				
107	2015-Oct	-15.32	2.76	-15.29	-11.52	-11.89	-11.60				
108	2015-Nov	-15.55	2.76	-15.52	-11.73	-12.10	-11.80				
109	2015-Dec	-15.78	2.76	-15.75	-11.94	-12.31	-12.00				
110	2016-Jan	-16.01	2.76	-15.98	-12.15	-12.52	-12.20				
111	2016-Feb	-16.24	2.76	-16.21	-12.36	-12.73	-12.40				
112	2016-Mar	-16.47	2.76	-16.44	-12.57	-12.94	-12.60				
113	2016-Apr	-16.70	2.76	-16.67	-12.78	-13.15	-12.80				
114	2016-May	-16.93	2.76	-16.90	-12.99	-13.36	-13.00				
115	2016-Jun	-17.16	2.76	-17.13	-13.20	-13.57	-13.20				
116	2016-Jul	-17.39	2.76	-17.36	-13.41	-13.78	-13.40				
117	2016-Aug	-17.62	2.76	-17.59	-13.62	-13.99	-13.60				
118	2016-Sep	-17.85	2.76	-17.82	-13.83	-14.20	-13.80				
119	2016-Oct	-18.08	2.76	-18.05	-14.04	-14.41	-14.00				
120	2016-Nov	-18.31	2.76	-18.28	-14.25	-14.62	-14.20				
121	2016-Dec	-18.54	2.76	-18.51	-14.46	-14.83	-14.40				
122	2017-Jan	-18.77	2.76	-18.74	-14.67	-15.04	-14.60				
123	2017-Feb	-19.00	2.76	-18.97	-14.88	-15.25	-14.80				
124	2017-Mar	-19.23	2.76	-19.20	-15.09	-15.46	-15.00				
125	2017-Apr	-19.46	2.76	-19.43	-15.30	-15.67	-15.20				
126	2017-May	-19.69	2.76	-19.66	-15.51	-15.88	-15.40				
127	2017-Jun	-19.92	2.76	-19.89	-15.72	-16.09	-15.60				
128	2017-Jul	-20.15	2.76	-20.12	-15.93	-16.30	-15.80				
129	2017-Aug	-20.38	2.76	-20.35	-16.14	-16.51	-16.00				
130	2017-Sep	-20.61	2.76	-20.58	-16.35	-16.72	-16.20				
131	2017-Oct	-20.84	2.76	-20.81	-	-	-				

Crecimiento medio del IMAE

A diciembre de cada año

Año	Guatemala	El Salvador
2019	4.00	2.53
2020	-2.01	-8.48
2021	8.18	10.02

Consejo N° 9: Especifiquemos fechas como números



Frecuentemente en nuestras hojas de cálculo trabajamos con series de tiempo, es decir, con variables que están indexadas con fechas o periodos. En estos casos es recomendable **guardar** las fechas y periodos como una fecha de Excel, y **mostrarlas** en un formato apropiado utilizando **Home → Number → Number Format**.

Veamos de vuelta el ejemplo de la figura 8: a pesar de que en la fórmula hemos buscado el valor del año "2019" (celda **AA17**) en la columna "Mes", hemos logrado que coincida con "2019-Dec" (celda **S181**). En realidad, en ambos caso las dos fechas están almacenadas como el número entero 43800, que corresponde al 1 de diciembre de 2019, pero en un caso tiene formato "yyyy" (año) y en el otro "yyyy-mm" (mes).

Trabajar con fechas como números (en vez de textos, como "2019/diciembre") tiene la enorme ventaja de que permite utilizar fórmulas de Excel para manipular fechas. Por ejemplo, supongamos que necesitamos calcular las exportaciones acumuladas por año, como en la figura 9a. Si lo hiciéramos con rangos de celdas, tendríamos que tener el cuidado de cambiar el rango de fechas en cada año de la tabla. Otra vez **OFFSET** al rescate: sumamos, a partir de la fila actual de la columna de exportaciones **[@exportaciones]**, tantas celdas hacia arriba como trimestres hayan transcurrido **-MONTH([@Mes])/3**:

=SUM(OFFSET([@exportaciones], 0, 0, -MONTH([@Mes])/3,1))

Para que esto funcionara correctamente, previamente hemos identificado los trimestres en la columna A con **fechas**: en este caso con el último día del trimestre respectivo. En la figura 9b vemos en la columna A como Excel efectivamente guarda esas fechas como números enteros. Trabajar con fechas también es conveniente a la hora de hacer gráficos: si en uno de los ejes debemos representar el tiempo, Excel podrá escoger una representación apropiada de los períodos si éstos están almacenados como fechas.

	A	B	C
1	Mes	exportaciones	acumuladas
2	2020-Mar	60	60
3	2020-Jun	50	110
4	2020-Sep	70	180
5	2020-Dec	80	260
6	2021-Mar	40	40
7	2021-Jun	70	110
8	2021-Sep	20	130
9	2021-Dec	80	210
10	2022-Mar	90	90
11	2022-Jun	50	140

(a) Los valores

	A	B	C
1	Mes	exportaciones	acumuladas
2	43921	60	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
3	44012	50	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
4	44104	70	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
5	44196	80	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
6	44286	40	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
7	44377	70	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
8	44469	20	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
9	44561	80	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
10	44651	90	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))
11	44742	50	=SUM(OFFSET([@exportaciones],0,0,-MONTH([@Mes])/3,1))

(b) Las fórmulas

Figura 9: Fórmulas que incluyen fechas

Consejo N° 10: Compartamos datos con Power Query

Muchos de los trabajos que realizamos en el día a día consisten en calcular estadísticas, a partir de datos que están almacenados en archivos de Excel de nuestros colegas, y posiblemente alguien más también necesitará leer nuestros datos calculados como insumos.

La manera más usual en que esto se realiza es mediante referencias a celdas en los archivos de Excel de otros compañeros. Esto tiene varios inconvenientes: (1) como mencionamos en el consejo 1, al hacer referencia a rangos de celdas como \$AX18:\$BD95, es difícil imaginar qué datos están contenidos allí, lo que nos obliga a abrir ese archivo (que podría estar bloqueado por otro usuario); (2) al abrir nuestro archivo, Excel intentará actualizar todas esas referencias, posiblemente bloqueándose durante minutos si son muchas las celdas vinculadas; y pero aún, (3) al no lograr leer los datos arrojará errores de referencia **#REF!** en todas esas celdas; (4) si logra actualizar los datos, podríamos entonces tener datos incorrectos, posiblemente porque aún estaban siendo procesados por nuestro colega, en cuyo caso no debimos haber actualizado aún nuestras celdas vinculadas.



Acá es donde sacamos mayor provecho del consejo 3: si los datos están organizados como tablas, entonces resulta muy sencillo compartir esas tablas con **Power Query**, así como controlar cuándo se actualizan esos datos en nuestros archivos. Power Query está incluido en Excel, aunque quizá es una herramienta poco conocida. A pesar de ello, resulta ser una herramienta muy sencilla de utilizar.

Por ejemplo, supongamos que uno de nuestros compañeros tiene un archivo en el cual ha calculado las tasas de variación interanual del IMAE de los países de CARD, y que deseamos leer esos datos, tal vez solo para hacer algunos gráficos con ellos. Para esta tarea, vamos a **Data → Get & Transform Data → Get Data** (ver figura 10) y escogemos **From File – From Excel Workbook**. Una vez escogido el archivo, veremos una ventana como la de la figura 11.





Figura 10: Importar datos

Del lado izquierdo de la ventana Navigator, vemos un listado de todas las tablas, hojas, y rangos definidos con nombre contenidos en el archivo (Funciones LAMBDA.xlsx, en nuestro ejemplo). Al seleccionar la tabla IMAE_VI, del lado derecho vemos su contenido (los valores null simplemente significan que para esas fechas específicas no hay datos). Para importar los datos a nuestro archivo de trabajo actual, simplemente hacemos clic en el botón **Load** y Excel hará una nueva hoja y colocará la tabla en la celda A1.

A partir de esta copia de la tabla podemos hacer nuestros propios cálculos y figuras. Y cada vez que necesitemos actualizar estos datos, seleccionaremos cualquier celda de la tabla y

luego **Query** → **Load** → **Refresh**.



The screenshot shows the Power Query interface. On the left, the Navigator pane displays a list of files under 'Funciones LAMBDA.xlsx [34]'. The file 'IMAE_VI' is selected and highlighted in green. The main view displays the 'IMAE_VI' table, which contains data for the months of 2007 across six countries: Costa Rica, El Salvador, Guatemala, Honduras, and Nicaragua. The data is presented in a table format with 'Mes' as the first column and the country names as subsequent columns. The data for all cells is currently 'null'.

Mes	Costa Rica	El Salvador	Guatemala	Honduras	Nicaragua
2007-01-01	null	null	null	null	null
2007-02-01	null	null	null	null	null
2007-03-01	null	null	null	null	null
2007-04-01	null	null	null	null	null
2007-05-01	null	null	null	null	null
2007-06-01	null	null	null	null	null
2007-07-01	null	null	null	null	null
2007-08-01	null	null	null	null	null
2007-09-01	null	null	null	null	null

At the bottom of the interface, there are three buttons: 'Load', 'Transform Data', and 'Cancel'.


Figura 11: Importando una tabla de otro archivo con Power Query

Consejo N° 11: Transformemos datos con Power Query

Ahora bien, es posible que en la práctica no necesitéramos importar todos los datos de una tabla, sino apenas una muestra. En vez de filtrar los datos en Excel directamente (por ejemplo, como **XLOOKUP** como en el consejo 8), haremos uso de Power Query para entonces leer solamente los datos que efectivamente necesitamos. Lo que es mejor, Power Query nos permite transformar, agrupar, e incluso crear nuevas columnas en la consulta misma, con lo cual facilitamos enormemente la labor.

Para ilustrar las posibilidades, supongamos ahora que el motivo por el que queremos importar esos datos es porque necesitamos elaborar un gráfico que compare el promedio trimestral de la tasa de variación interanual del IMAE de Guatemala y República Dominicana desde 2018. En tal caso, cuando estamos seleccionando los datos (figura 11), en vez de **Load** presionamos **Transform Data** y preparamos los datos que necesitamos.

Primero, filtramos los datos para tener solo observaciones posteriores a 2017: seleccionamos la pequeña flecha en la columna Mes (ver figura 12a), luego **Date Filters** y **After**, con lo cual tenemos una ventana como en la figura 12b, en la cual seleccionamos las fechas deseadas.




(a) Filtrando los datos

(b) Escogiendo las fechas


Figura 12: Simplificando nuestros cálculos con funciones LAMBDA

Segundo, transformamos la columna de "Mes" a trimestres: para ello, seleccionamos esa columna y con el botón secundario del ratón se despliega el menú de la figura 13a, en el cual seleccionamos **Transform → Quarter → End of Quarter**. Ahora todos los datos de la columna Mes muestran el último día del trimestre: vemos que cada dato aparece repetido tres veces (los tres meses de cada trimestre).

Tercero, calculamos los promedios trimestrales para Guatemala y República Dominicana, empezando con **Transform → Table → Group By** (figura 13b). Esto abrirá la ventana de la



(a) Filtrando los datos



(b) Escogiendo las fechas

Figura 13: Simplificando nuestros cálculos con funciones LAMBDA

figura 14a, en la cual hemos indicado que queremos agrupar los datos por mes, creando dos columnas: GT, con el promedio de los datos de Guatemala, y DO, con los de República Dominicana. Aunque hemos usado los códigos ISO de ambos países, no hay ningún inconveniente si simplemente deseamos usar los nombres originales que tenían en la tabla de datos originales.

Group By

Specify the columns to group by and one or more outputs.

☐ Basic ☒ Advanced

Mes

Add grouping

New column name

GT

Operation

Average

Column

Guatemala

DO

Average

República Dominicana

Add aggregation

OK

Cancel

	A	B	C
1	Mes	GT	DO
2	2018-03	2.81	6.95
3	2018-06	4.31	7.63
4	2018-09	3.72	7.11
5	2018-12	3.24	5.75
6	2019-03	3.68	5.24
7	2019-06	3.98	4.75
8	2019-09	4.02	5.13
9	2019-12	4.33	4.12
10	2020-03	-0.97	-3.89
11	2020-06	-6.76	-9.97
12	2020-09	-1.96	-7.75
13	2020-12	1.68	-2.50
14	2021-03	6.52	7.90
15	2021-06	13.29	15.75
16	2021-09	8.46	12.84
17	2021-12	4.87	9.34
18	2022-03	4.65	6.61
19	2022-06	3.88	5.85
20			


(a) Definiendo las agrupaciones

(b) Tabla importada


Figura 14: Simplificando nuestros cálculos con funciones LAMBDA

Una vez que hemos terminado de transformar los datos, estamos listos para cargarlos en nuestro archivo de Excel. Para ello en la ventana de Power Query vamos a **Home → Close**

→ **Close & Load** (ver figura 15a). Al escoger la opción "Close & Load To..." se despliega la ventana de la figura 15b, donde podemos escoger el lugar donde se guardará la tabla, o bien utilizarla como base para una tabla dinámica.



(a) Cargando los datos



(b) Escogiendo el destino

Figura 15: Simplificando nuestros cálculos con funciones LAMBDA

El resultado final aparece en la figura 14b, en la cual vemos los datos descargados, luego de haberles cambiado el formato usando las herramientas usuales.

Consejo N° 12: Complementemos Excel con otras herramientas

El último de los consejos en este documento está más podría considerarse el consejo "anti-Excel": aprendamos a utilizar herramientas alternativas para el procesamiento de datos.

La principal fortaleza de Excel es que nos permite trabar con datos de manera interactiva, de forma tal que podemos explorar paso a paso las distintas transformaciones que debemos implementar, y terminamos por enlazar en celdas todo un flujo de información.

Pero esta fortaleza es también, a mi parecer, una enorme debilidad. Al ser Excel tan flexible, es sumamente fácil hacer hojas de cálculo desordenadas, donde es difícil entender rápidamente el flujo de información que está ocurriendo en un archivo de trabajo. Además, al estar las celdas vinculadas, Excel recalcula constantemente sus valores cada vez que modificamos una celda, lo cual puede ser muy lento cuando se trabaja con cantidades grandes de datos.

Además, una vez que se han escrito correctamente todas las fórmulas de un archivo, ya no es tan necesario la interactividad del libro de Excel. En tales caso, lo más conveniente es que estas transformaciones a los datos se automaticen con un lenguaje de programación.

Es por ello que cierro esta lista de consejos con una exhortación: **¡Aprendamos Python o R!** Estas herramientas están especialmente diseñadas para manipular datos, y al trabajar por medio de archivos de código ("scripts") que se ejecutan de manera secuencial, es sumamente fácil de auditar. Además, puesto que en estos lenguajes se puede manipular muchísimos datos con una sola fórmula (distinto de lo que usualmente se hace en Excel, donde cada celda-dato tiene su propia fórmula), es más fácil detectar errores: cuando en Python o R una fórmula está mal, todos los resultados estarán mal, lo cual levanta una sospecha inmediata; en Excel, si algunas de las miles de fórmulas de un libro están mal, es mucho más difícil de detectar.

Finalmente, estos programas alternativos pueden ser un excelente complemento para Excel. Por ejemplo, en Python los datos de Excel pueden leerse y transformarse con librerías como **pandas**, y luego puede crearse todo un nuevo archivo de Excel (con formato y todo) con la librería **openpyxl**. Este flujo de trabajo es especialmente recomendable para labores periódicas, como el cálculo de estadísticas económicas mensuales o bien la elaboración de informes.

Esto lo ilustramos en la figura 16 donde vemos dos archivos abiertos: a la izquierda un libro de Excel tradicional, en el cual se combinan datos "crudos" o de "insumo" con datos obtenidos con fórmulas; y a la derecha un *script* o archivo de instrucciones para Python. En los consejos anteriores hemos visto cómo evitar fórmulas tan engorrosas como la de la celda **H362** (que a no-tan-simple-vista vemos que calcula la variación acumulada media de

la serie que está en la columna F), organizando los datos en tablas y definiendo funciones con `LAMBDA`, pero aún tenemos el inconveniente de que debemos copiar manualmente las estas funciones a cada archivo de trabajo en el cual las necesitemos (en el apéndice veremos cómo se hace esto). La alternativa que ilustra la parte derecha de la figura 16 asume que los datos "insumo" han sido debidamente almacenados en un libro de Excel llamado `precios.xlsx`, el cual leemos (línea 15) con la función `pd.read_excel`. Creamos un nuevo archivo de Excel en la línea 17, al que llamaremos `resultados.xlsx`, al cual le agregamos dos hojas: "VI", con la variación interanual de los datos (línea 19) y "VA" con la variación acumulada (línea 20). Una vez que terminamos nuestros cálculos, cerramos el archivo de Excel que hemos creado (línea 22).

¿Cómo sabe Python calcular variaciones interanuales y acumuladas? Para ello hemos de antemano definido funciones que hacen esos cálculos, de manera similar a cuando hicimos las funciones lambda de Excel, y las hemos almacenado en un módulo (un archivo) al que hemos llamado `miPaquete`. Para utilizarlas, simplemente le decimos a Python en la línea 13 que las "importe" en nuestro archivo de trabajo.

El archivo de Python de la figura 16 ilustra también algo muy importante: si desarrollamos bien nuestro código, el mismo se convierte en la *documentación* de nuestro trabajo, porque es muy sencillo entender todo el proceso que hemos realizado: acá simplemente leemos datos de Excel (posiblemente los niveles de varios índices de precios), calculamos sus tasas de variación interanual y acumulada, y guardamos los resultados en otro archivo de Excel. ¡Así de sencillo! Trabajando solamente con Excel, tendríamos que ver las fórmulas de muchas celdas para deducir el flujo de información que allí ocurre.




Figura 16: Complementando Excel con Python

Apéndice

A Ejemplos de funciones LAMBDA

En esta sección comparto algunos ejemplos de funciones **LAMBDA** que podrían resultar de utilidad para el manejo de series de tiempo. En todos los casos, se asume que los datos se han organizado como una tabla, donde cada fila representa un mes. Se deja como ejercicio para el lector hacer funciones nuevas para series de tiempo de otra frecuencia, por ejemplo trimestrales.

Excel aún no ha desarrollado un mecanismo apropiado para compartir funciones lambda entre distintos archivos. Por lo pronto, hay dos maneras para utilizar estas nuevas funciones en sus libros de trabajo. La primera es copiar las funciones en el cuadro de nombres, con **Formulas → Name Manager → Define Name**, tal como lo vimos en el consejo 7, teniendo cuidado de copiar además las fórmulas que estén referidas en la que se necesita (por ejemplo, para usar `sumar_valores`, necesita copiar también `no_faltan_datos`). La segunda manera, mucho más sencilla, es copiar cualquier hoja del libro de Excel que acompaña esta nota (descargable con el icono de al lado) al libro de trabajo donde necesite las fórmulas; hecho esto, todas las fórmulas estarán disponibles en ese libro, y podrá borrar la copia de la hoja que colocó en ese archivo.



Las funciones lambda disponibles son las siguientes:

no_faltan_datos	21
sumar_valores	21
promediar_valores	21
numero_valido	21
cambio_porcentual	22
rezago	22
suma_acumulada_este_año	22
suma_acumulada_año_pasado	23
promedio_acumulado_este_año	23
variacion_acumulada	23
variacion_mensual	24
variacion_mensual_anualizada	24
variacion_interanual	24
diferencia	24
media_movil	25
crecimiento_ultimo_año	25



Documento de Trabajo

Secretaría Ejecutiva del Consejo Monetario Centroamericano

No. 2, junio de 2022

no_faltan_datos

```
=LAMBDA(rango,  
        AND(ISNUMBER(rango)))
```

Verifica que todas las celdas de un rango de datos contengan valores numéricos. Necesario porque Excel al hacer cálculos toma celdas vacías como si fuesen ceros o los omite de los cálculos.

rango	$\{x_1, x_2, \dots, x_n\}$	un rango de celdas, con o sin valores
-------	----------------------------	---------------------------------------

$\text{no_faltan_datos}(\text{rango}) = 1$	si $x_j \in \mathbb{R} \quad \forall j \in \{1, 2, \dots, n\}$, de lo contrario 0.
--	---

sumar_valores

```
=LAMBDA(rango,  
        IF(no_faltan_datos(rango), SUM(rango), ""))
```

Verifica que todas las celdas de un rango de datos contengan valores numéricos. De ser así, devuelve la suma de esos valores, caso contrario devuelve "" (es decir, deja la celda en blanco).

rango	$\{x_1, x_2, \dots, x_n\}$	un rango de celdas, con o sin valores
-------	----------------------------	---------------------------------------

$\text{sumar_valores}(\text{rango}) = \sum_{j=1}^n x_j$	si $x_j \in \mathbb{R} \quad \forall j \in \{1, 2, \dots, n\}$, de lo contrario \emptyset .
--	--

promediar_valores

```
=LAMBDA(rango,  
        IF(no_faltan_datos(rango), AVERAGE(rango), ""))
```

Verifica que todas las celdas de un rango de datos contengan valores numéricos. De ser así, devuelve el promedio de esos valores, caso contrario devuelve "" (es decir, deja la celda en blanco).

rango	$\{x_1, x_2, \dots, x_n\}$	un rango de celdas, con o sin valores
-------	----------------------------	---------------------------------------

$\text{promediar_valores}(\text{rango}) = \frac{1}{n} \sum_{j=1}^n x_j$	si $x_j \in \mathbb{R} \quad \forall j \in \{1, 2, \dots, n\}$, de lo contrario \emptyset .
--	--

numero_valido

```
=LAMBDA(x,  
        IF(ISNUMBER(x), x, ""))
```




Documento de Trabajo

Secretaría Ejecutiva del Consejo Monetario Centroamericano

No. 2, junio de 2022

Si "x" es un número, devuelve ese número. Caso contrario devuelve "" (es decir, deja la celda en blanco).

x x una celda o expresión, con o sin valor

`numero_valido(x) = x` si $x \in \mathbb{R}$, de lo contrario \emptyset .

cambio_porcentual

```
=LAMBDA(inicial, final,  
         numero_valido(100*(final/inicial-1)))
```

Calcula el cambio porcentual entre dos valores:

inicial y_{t-s} valor inicial
final y_t valor final

`cambio_porcentual(inicial, final) = 100 $\left(\frac{y_t}{y_{t-s}} - 1 \right)$`

rezago

```
=LAMBDA(serie, periodos,  
         LET(n, IF(periodos, periodos,1),  
             x, OFFSET(serie,-n,0),  
             numero_valido(x)))
```

Calcula el valor rezagado de una "serie"

serie y_t valor de la serie
periodos s número entero, opcional (1)

`rezago(serie, periodos) = y_{t-s}`

`rezago(serie,) = y_{t-1}` (asume `periodos=1`. Nota: la coma "," es necesaria.)

suma_acumulada_este_año

```
=LAMBDA(serie, fecha,  
         LET(valores, OFFSET(serie,0,0,-MONTH(fecha),1),  
             sumar_valores(valores)))
```

Suma acumulada de una serie mensual, de este año.

serie $y_{t,m}$ valor de la serie en mes m del año t
fecha t, m fecha de Excel (número entero)

`suma_acumulada_este_año(serie, fecha) = $\sum_{j=1}^m y_{t,j}$`

suma_acumulada_año_pasado

```
=LAMBDA(serie, fecha,
    LET(valores, OFFSET(serie,-12,0,-MONTH(fecha),1),
        sumar_valores(valores)))
```

Suma valores mensuales de "serie" acumulados entre enero y el mes de "fecha" pero del año anterior.

serie	$y_{t,m}$	valor de la serie en mes m del año t
fecha	t, m	fecha de Excel (número entero)

$$\text{suma_acumulada_este_año}(\text{serie}, \text{fecha}) = \sum_{j=1}^m y_{t-1,j}$$

promedio_acumulado_este_año

```
=LAMBDA(serie, fechas,
    LET(x, OFFSET(serie,1-MONTH(fechas),0,MONTH(fechas),1),
        promediar_valores(x)))
```

Calcula el promedio de "serie" entre enero y el mes actual.

serie	$y_{t,m}$	valor de la serie en mes m del año t
fecha	t, m	fecha de Excel (número entero)

$$\text{promedio_acumulado_este_año}(\text{serie}, \text{fecha}) = \frac{1}{m} \sum_{j=1}^m y_{t,j}$$

variacion_acumulada

```
=LAMBDA(serie, fecha,
    LET(inicial, suma_acumulada_año_pasado(serie,fecha),
        final, suma_acumulada_este_año(serie,fecha),
        cambio_porcentual(inicial, final)))
```

Calcula la variación acumulada media de un año respecto al año anterior.

serie	$y_{t,m}$	valor de la serie en mes m del año t
fecha	t, m	fecha de Excel (número entero)

$$\text{variacion_acumulada}(\text{serie}, \text{fecha}) = 100 \left(\frac{\sum_{j=1}^m y_{t,j}}{\sum_{j=1}^m y_{t-1,j}} - 1 \right)$$



Documento de Trabajo

Secretaría Ejecutiva del Consejo Monetario Centroamericano

No. 2, junio de 2022

variacion_mensual

```
=LAMBDA(serie,  
        LET(inicial, rezago(serie,1),  
            cambio_porcentual(inicial, serie)))
```

serie	y_t	valor de la serie en t
-------	-------	--------------------------

$$\text{variacion_mensual(serie)} = 100 \left(\frac{y_t}{y_{t-1}} - 1 \right)$$

variacion_mensual_anualizada

```
=LAMBDA(serie,  
        LET(inicial, rezago(serie,1),  
            numero_valido(100*((serie/inicial)^12-1))))
```

serie	y_t	valor de la serie en t
-------	-------	--------------------------

$$\text{variacion_mensual_anualizada(serie)} = 100 \left[\left(\frac{y_t}{y_{t-1}} \right)^{12} - 1 \right]$$

variacion_interanual

```
=LAMBDA(serie,  
        LET(inicial, rezago(serie,12),  
            cambio_porcentual(inicial,serie)))
```

Calcula la variación interanual de una serie.

serie	y_t	valor de la serie en t
-------	-------	--------------------------

$$\text{variacion_interanual(serie)} = 100 \left(\frac{y_t}{y_{t-12}} - 1 \right)$$

diferencia

```
=LAMBDA(serie, periodos,  
        LET(inicial, rezago(serie, periodos),  
            numero_valido(serie-inicial)))
```

serie	y_t	valor de la serie
periodos	s	número entero, opcional (1)

$$\text{diferencia(serie, periodos)} = y_t - y_{t-s}$$

$$\text{diferencia(serie,)} = y_t - y_{t-1} \text{ (asume periodos= 1. Nota: la coma ", " es necesaria.)}$$



Documento de Trabajo

Secretaría Ejecutiva del Consejo Monetario Centroamericano

No. 2, junio de 2022

media_movil

```
=LAMBDA(serie, periodos,  
        LET(x, OFFSET(serie,0,0,-periodos,1),  
            promediar_valores(x)))
```

Calcula el promedio móvil de los últimos valores de una serie.

serie	y_t	valor de la serie
periodos	s	número entero, opcional (1)

$$\text{media_movil}(\text{serie}, \text{periodos}) = \frac{1}{s+1} \sum_{j=0}^s y_{t-j}$$

crecimiento_ultimo_año

```
=LAMBDA(serie,  
        LET(anterior, sumar_valores(OFFSET(serie,-12,0,-12,1)),  
            actual, sumar_valores(OFFSET(serie,0,0,-12,1)),  
            cambio_porcentual(anterior,actual)))
```

Calcula la tasa de crecimiento de los últimos doce meses con respecto a los doce meses anteriores

serie	y_t	valor de la serie
-------	-------	-------------------

$$\text{crecimiento_ultimo_año}(\text{serie}, \text{fecha}) = 100 \left(\frac{\sum_{j=1}^{12} y_{t+1-j}}{\sum_{j=13}^{24} y_{t+1-j}} - 1 \right)$$

B Algunas funciones de Excel

En el siguiente cuadro están todas las funciones de Excel utilizadas a lo largo de este documento, así como su equivalente en español y una pequeña descripción.

Función	En español	Descripción
AND	Y	Determina si todas las condiciones de una prueba son VERDADERAS.
AVERAGE	PROMEDIO	Devuelve el promedio (media aritmética) de los argumentos.
IF	SI	Evalúa una condición y devuelve uno de dos resultados, dependiendo si la condición es verdadera o falsa.
ISNUMBER	ESNUMERO	Devuelve el valor lógico verdadero si el valor de la celda se refiere a un número.
LAMBDA	LAMBDA	Crea funciones personalizadas reutilizables que pueden ser llamadas por un nombre descriptivo.
LET	LET	Asigna nombres a los resultados de cálculos intermedios, para utilizarse posteriormente en la misma fórmula.
MONTH	MES	Devuelve el mes de una fecha representada por un número de serie, expresado como número entero comprendido entre 1 (enero) y 12 (diciembre).
OFFSET	DESREF	Devuelve una referencia a un rango que es un número de filas y de columnas de una celda o rango de celdas.
SUM	SUMA	Devuelve la suma valores, estén estos en celdas individuales, rangos o referencias, o una combinación de las tres.
VLOOKUP	BUSCARV	Busca elementos en una tabla o un rango por fila.
XLOOKUP	BUSCARX	Busca elementos en una tabla o rango por fila.

Cuadro 1: Funciones de Excel utilizadas en este artículo, y sus equivalentes en Excel en español

Referencias

- Alexander, Michael y Dick Kusleika (2022). *Microsoft Excel 365 Bible*. Wiley. ISBN: 978-1-119-83510-3.
- Borwein, Jonathan y David H. Bailey (22 de abr. de 2013). *The Reinhart-Rogoff error – or how not to Excel at economics*. Ed. por The Conversation. URL: <https://theconversation.com/the-reinhart-rogoﬀ-error-or-how-not-to-excel-at-economics-13646>.
- Herndon, Thomas, Michael Ash y Rober Pollin (2013). *Does High Public Debt Consistently Stifle Economic Growth? A Critique of Reinhart and Rogoff*. Inf. téc. 322. Political Economy Research Institute, University of Massachussetts Amherst.
- Hurtado, Patricia (23 de feb. de 2016). *The London Whale*. Ed. por Bloomberg. URL: <https://www.bloomberg.com/quicktake/the-london-whale>.
- Lopez, Linette (12 de feb. de 2013). *How The London Whale Debacle Is Partly The Result Of An Error Using Excel*. Ed. por Business Insider. URL: <https://www.businessinsider.com/excel-partly-to-blame-for-trading-loss-2013-2>.
- Microsoft (20 de jun. de 2022a). *Ayuda y formación de Excel*. URL: <https://support.microsoft.com/es-es/excel>.
- (20 de jun. de 2022b). *Excel help & learning*. URL: <https://support.microsoft.com/en-us/excel>.
- Reinhart, Carmen M. y Kenneth S. Rogoff (2010). *Growth in a Time of Debt*. Inf. téc. 15639. NBER.

